



● 产品名称

➤ MK7A11P

● 标题

如何使用 MK7A11P 的 TMR0

● 简介

MK7A11P 的 TMR0 是一个 8 位递减计数器，是由 SELECT 寄存器来控制的，SELECT 寄存器如下图所示：

Name	Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0
SELECT	TMR0_EN	AUTO	SUR0	EDGE0	PSA	PS2	PS1	PS0

Bit	Symbol	Description				
2~0	PS2~PS0	PS2	PS1	PS0	TMR0 rate	WDT rate
		0	0	0	1:2	1:1
		0	0	1	1:4	1:2
		0	1	0	1:8	1:4
		0	1	1	1:16	1:8
		1	0	0	1:32	1:16
		1	0	1	1:64	1:32
		1	1	0	1:128	1:64
		1	1	1	1:256	1:128
3	PSA	PSA: Prescaler assignment bit 1: Prescaler assigned to WDT 0: Prescaler assigned to TMR0				
4	EDGE0	EDGE0: TMR0 source signal edge control bit 1: increment when H→L transition on external clock 0: increment when L→H transition on external clock				
5	SUR0	SUR0: TMR0 clock source bit 1: External clock input 0: (Internal clock) / 4				
6	AUTO	AUTO : Auto pre-load TMR0 data 1: enable 0: disable				
7	TMR0_EN	TMR0_EN: TMR0 enable/disable 0: disable 1: enable				

向 SELECT 寄存器输入数据必须通过执行 select 指令来完成。例如：

```
movla    b'11000010'  
select
```



上述指令执行完后，SELECT 寄存器里面的值将为 b'11000010'，这样 TMR0 的配置结果如下：

- 预分频比：1：8
- 时钟源：(系统时钟)/4
- TMR0 开启

为了让用户更深入地了解 MK7A11P 的 TMR0 的用法，我们编写一个让 MK7A11P 的 PA0 口产生一个周期为 2s，占空比为 50% 方波的程序，以供用户参考。

● DEMO 程序

➤ 汇编程序文档

```
;-----  
#include "mk7a11p_hw.inc" ;编译该文档需包含"mk7a11p_hw.inc"文件  
;-----  
;芯片型号 (mk7a11p)  
;-----  
;配置寄存器设置说明 (CONFIG)  
;1-----FOSC=RC ;LS,NS,HS,RC  
;2-----INRC=ON ;ON,OFF  
;3-----CPT=OFF ;ON,OFF  
;4-----WDTE=Disable ;Enable,Disable  
;5-----LV=Low Vol Reset ON ;Low Vol Reset ON,Low Vol Reset OFF  
;6-----RESET=...input... ;...input...,...reset...  
;-----  
tm_r equ 0x20 ;1s 计时  
;-----  
org 0x3ff ;mk7a11p 的复位向量地址定义  
lgoto main ;跳转到主程序入口  
;-----  
org 0x100 ;主程序入口地址定义  
main  
;PortA 端口方向及状态设定  
movla b'11110000'  
iodir porta  
clr porta  
clr pa_pdm  
;-----  
;PortB 端口方向及状态设定  
movla b'11111111'  
iodir portb
```



```
clr                portb
clr                pb_pod
clr                pb_pdm
clr                pb_pup
;-----
;PortB 端口 8 个 IO 唤醒使能
;movla            b'11111111'
;movam            wake_up
;-----
;配置 TMR0,预分频比为 1:16
;TMR0 初始值为 250
movla              b'11000011'
select
movla              .250
movam              tmr0
;-----
;中断禁止
clr                irqm
clr                irqf
;-----
loop                ;程序循环入口
;程序循环周期控制处 (T=4ms)
btss               irqf,0
lgoto              $-1                ;等待 TMR0 溢出
bc                 irqf,0            ;清除溢出标志
;-----
inc                tm_r,m
movla              .250                ;4x250=1s
xor                tm_r,a
btss               status,z
lgoto              loop
clr                tm_r
;-----
movla              b'00000001'
xor                porta,m            ;PA0 的 High Low 切换
lgoto              loop
;-----
end
➤ mk7a11p_hw.inc 文档
;-----Define special register(Define SFR) -----
```



indf	equ	0x00	
tmr0	equ	0x01	
pc	equ	0x02	
status	equ	0x03	
fsr	equ	0x04	
porta	equ	0x05	;porta(0-3)
portb	equ	0x06	;portb(0-7)
;-----			
irqm	equ	0x09	
irqf	equ	0x0a	
;-----			
pa_pdm	equ	0x0b	
pb_pup	equ	0x0c	
pb_pdm	equ	0x0d	
pb_pod	equ	0x0e	
wake_up	equ	0x0f	
;-----Define [status Register] special bit-----			
c	equ	0	
dc	equ	1	
z	equ	2	
pd	equ	3	
to	equ	4	
;-----Define [irqm Register] special bit-----			
tm0m	equ	0	
extm	equ	1	
intm	equ	7	
;-----Define [irqf Register] special bit-----			
tm0f	equ	0	
extf	equ	1	
;-----Define [pa_pdm Register] special bit-----			
da0	equ	0	
da1	equ	1	
da2	equ	2	
da3	equ	3	
;-----Define [pb_pup Register] special bit-----			
ub0	equ	0	
ub1	equ	1	
ub2	equ	2	
ub4	equ	4	



ub5	equ	5
ub6	equ	6
ub7	equ	7
;-----Define [pb_pdm Register] special bit-----		
db0	equ	0
db1	equ	1
db2	equ	2
inte	equ	6
rtce	equ	7
;-----Define [pb_pod Register] special bit-----		
ob0	equ	0
ob1	equ	1
ob2	equ	2
ob4	equ	4
ob5	equ	5
ob6	equ	6
ob7	equ	7
;-----Define [wake_up Register] special bit-----		
en0	equ	0
en1	equ	1
en2	equ	2
en3	equ	3
en4	equ	4
en5	equ	5
en6	equ	6
en7	equ	7
;-----		